
Ruim een jaar geleden hebben we goedkoop een 8280 gekocht en door alle leuke dingen 
die deze computer meer kan dan een ‘gewone’ MSX, zijn we er heel erg gelukkig mee. 
Zo gelukkig, dat we hem ook mee naar Japan namen waar we voor voor de studie heen 
moesten. Daar hebben hem bijna niet kunnen uittesten, want zoals sommigen wel weten, 
gebruikt men daar NTSC. Meer dan een zwart-wit beeldje krijg je dan ook niet. Maar 
gelukkig heeft ons MSX’je de reis naar Nederland met de PTT overleefd! De elektronica 
al-thans wel, want de kast hadden om veiligheidsredenen en kosten maar thuisgelaten. En 
bedenk: als je MSX de PTT overleeft, dan moet ’ie er ook tegenkunnen om van driehoog 
naar beneden te vallen, om vervolgens te worden overreden door een bende scootertuig.  
Na onze neuzen te hebben gestoken in de V9938–handleiding, bleek de aansturing toch 
wel wat ingewikkelder dan we dachten. Dus zijn we maar eens gaan hacken in Video-
Paint 4, waaruit na veel ploeteren de juiste instellingen naar boven kwamen. Ik heb ze in 
een basicpro-gramma gezet om het mogelijk te maken wat meer te spelen met de digitize-
functie. Alle superimpose-functies zijn wel terug te vinden in een basichandboek. Elke 
basicregel heeft wat commentaar voor als je het in machinetaal wilt gebruiken.  
 
Theorie 
 
We hebben in het programma ‘gewoon’ scherm 8 gebruikt. Daarmee gaat alles zo als het 
hoort — behalve de bruine kleuren, die geloof ik wel te verhelpen zijn — omdat het 
hiervoor allemaal is gemaakt. De MSX haalt, elke keer dat een puntje opgebouwd moet 
worden, de data van de color bus, een 8–bit datapoort aan de VDP, en zet die in het 
geheugen op de plaats die correspondeert met dat puntje. Deze data is gelijk aan het 
kleurformaat voor scherm 8, dus in één byte: RRRGGG BB. Je kunt trouwens met 
VDP(7) een kleurmasker erover een plakken. Voordat de colorbusdata  het geheugen in 
gaat, wordt die met een AND–operatie met deze waarde bewerkt. In ons programma staat 
deze waarde in regel 20, de variabele CC. Maken we die dus gelijk aan &B00011100, 
dan zien we alleen de groene kleuren in het plaatje. 
We kunnen echter ook naar een andere schermmode gaan. Dan zet de VDP nog steeds 
braaf de colorbusdata in het geheugen. Maar de bitjes die nu niet nodig zijn, worden wel 
genegeerd. Je kunt dit eens proberen door in regel 30 SCREEN 5 neer te zetten. Je krijgt 
dan wel een lijkend plaatje, maar de kleuren komen niet meer overeen. Immers, de vier 
bits in scherm 5 verwijzen naar het kleurpaletnummer en niet naar een RGB–waarde, 
terwijl er wel een stukje van die RGB–waarde wordt neergezet. Het zou een beetje te ver-
helpen zijn door met het kleurmasker van &B00000011 de stukken groen en rood te 



negeren en alleen blauwe kleuren door te laten. Als laatste stap moeten we dan het 
kleuren palet zo aanpassen, dat het overeenkomt met deze 4 waarden. Het is verder ook 
mogelijk om in screen 6 of 7 en hoogstwaarschijn-lijk ook in 9, 11 of 12 te digitizen. Dat 
wil niet zeggen, dat het mogelijk is om op deze manier in screen 12 in full color te 
digitizen. Ons MSX'je heeft niet de snelheid om naast het ophalen van de bytes, ook nog 
eens de YJK–codering erover te halen. Hoe de Sony HB1V digitizer het zo snel doet, 
weet ik niet pre-cies, maar hij schuift de data uit een IO–poortje door naar de VDP. Met 
een heel snel processortje die vier bytes inleest, de YJK–codering uitrekent en aan de 
andere kant deze vier bytes weer netjes één voor één op de colorbus aanreikt aan de VDP, 
zou het mogelijk moeten zijn om netjes in scherm 12 full color te samplen. Misschien is 
er een hardwarefanaat die eens zin heeft om hier naar te kijken. M’n MSX wil ik best als 
testexemplaar aanbieden — als ie de PTT heeft overleefd, kan dit er nog wel bij. Maar 
voorlopig zullen we het moeten doen met scherm 8. 
 
Programma 
 
Nog een aantal opmerkingen bij hetbasic programma: 
· De TIME–instructie wordt gebruikt om even te wachten tot de VDP klaar is met het 
digitizen van het huidige scherm. Het programma stopt of hervat het digitizen door op de 
eerste vuurknop van de muisjoystick in poort 1 te drukken.  Met muisknop 2 wordt het 
programma beëindigd. 
· Poort &HF7 schakelt de verschillende (video) signalen en lampjes in de computer. Door 
de commando’s zomaar in basic in te typen, kan het zijn dat u de controle over de VDP 
verliest. Maar niets gevreesd, met een ferme druk op de resetknop komt het overbekende 
MSX beeld weer tevoorschijn. 
· Het SET PAGE–commando wordt genegeerd, zodat het niet mogelijk is om op een niet–
zichtbare pagina te digitizen. 
 
Toepassingen 
 
Zo zou je bijvoorbeeld een regel 135kunnen maken, waarin je het plaatje wegschrijft. De 
mogelijkheden zijn vrijwel onbeperkt. Je zou bij-voorbeeld met een videocamera, gericht 
op een kleurig achtergrond, kunnen uitzoeken of er iemand staat, of in een bepaalde 
positie staat, door het digitizen te stoppen en bepaalde punten op hun kleur 



 
10    Digitize met NMS 8280  
20    CC=&B11111111                  'color masker : GGGRRRBB  
50    SCREEN 8  
90    GOSUB 1000                     'start digitizing  
100   IF STRIG(3)=-1 THEN GOTO 160 'einde 20  
110   IF STRIG(1)=-1 THEN GOSUB 1200 :GOTO 130  ' pauze  
120   GOTO 10  
130   IF STRIG(1)=-1 THEN 130 
140   IF STRIG(1)=-1 THEN GOSUB 1000 :GOTO 100  ' restart digitzing  
141   IF STRIG(3)=-1 THEN GOTO 160   'einde  
150   GOTO 140  
160   GOSUB 1200                    'stop digitizing 
170   COLOR 15,0,0 :SCREEN 0 :END 
1000  '*-*-*-*-*-*-*-*-*-*-*-*-*-*-*-*-*- start digitizing  
1050  COLOR 0,0,0  
1060  VDP(0 )=VDP(0 ) AND &HBF     'b6=0  reg0  (digitize off)  
1080  TIME=0  
1090  IF TIME<8 THEN 1090          'wachtlusje  
1130  OUT(&HF7),&B10011111         'b7=1,b6=0  poort F7 (mix)  
1140  VDP(10 )=VDP(10 ) AND &HDF OR &H10  'b5=0 ,b4=1 reg10  (mix) 
1150  VDP(9)=VDP(9) OR 32          'b5=1 reg9 kleur0 =kleur0  palet  
1160  VDP(7)=CC                  'voorachtergrond kleur>maskercolorbus 
1170  VDP(0 )=VDP(0 ) OR &H40      'b6=1 reg0  (digitize on)  
1180  RETURN  '  
1200  '*-*-*-*-*-*-*-*-*-*-*-*-*-*-*-*-*-*-*-*-*-*-*-*-*-*-* stop   
1210  VDP(0 )=VDP(0 ) AND &HBF     'b6=0  reg0  (digitizer off)  
1230  TIME=0  ' 
1240  IF TIME<2 THEN 1240          'wachtlusje  
1290  VDP(9)=VDP(9) AND 223        'b5=0  reg9 kleur0 =doorzichtig 
1300  OUT(&HF7),255                'b7=1:b6=1 (computer)  
1310  VDP(10)=VDP(10) AND &HEF     'b4=0  (computer),b5 was al 0 
1320  VDP(7)=0                'maskercolorbus->voorachtergrond kleur  
1330  RETURN '  
1400  '*-*-*-*-*-*-**-*-*-*-*-*-*-*- ' freeze  
1490  VDP(0 )=VDP(0 ) AND &HBF     'b6=0  reg0  (digitize off)  
1600 RETURN '  

